
ContextKit 
    



      
ContextKit provides a modular toolkit for building robust Phoenix/Ecto contexts with standardized CRUD operations.

  
    
  
  Overview


ContextKit aims to reduce boilerplate code in Phoenix applications by providing:
	Standardized CRUD operation generators
	Dynamic query building with extensive filtering options
	Built-in pagination support
	Support for Phoenix 1.8 scopes
	PubSub integration for real-time updates
	Flexible and extensible design


  
    
  
  Basic Usage


ContextKit offers two main modules:
	ContextKit.CRUD - Core CRUD operations with no scope support
	ContextKit.CRUD.Scoped - CRUD operations with Phoenix 1.8 scope support and PubSub integration

Choose the module that fits your needs:
	Use ContextKit.CRUD for simple contexts like in Phoenix 1.7 without scopes or PubSub needs
	Use ContextKit.CRUD.Scoped when working with Phoenix 1.8 scopes and wanting real-time updates


  
    
  
  Getting Started



  
    
  
  1. First, define your schema modules:


defmodule MyApp.Accounts.User do
  use Ecto.Schema
  import Ecto.Changeset

  schema "users" do
    field :email, :string
    field :name, :string
    field :status, :string

    has_many :comments, MyApp.Blog.Comment

    timestamps()
  end

  def changeset(user, attrs) do
    user
    |> cast(attrs, [:email, :name, :status])
    |> validate_required([:email, :name])
    |> unique_constraint(:email)
  end
end

defmodule MyApp.Blog.Comment do
  use Ecto.Schema
  import Ecto.Changeset

  schema "comments" do
    field :body, :string
    field :status, :string

    belongs_to :user, MyApp.Accounts.User

    timestamps()
  end

  # Standard changeset for unscoped operations
  def changeset(comment, attrs) do
    comment
    |> cast(attrs, [:body, :status, :user_id])
    |> validate_required([:body, :user_id])
  end

  # Scoped changeset for Phoenix 1.8 scope operations
  def changeset(comment, attrs, scope) do
    comment
    |> cast(attrs, [:body, :status])
    |> validate_required([:body])
    |> put_change(:user_id, scope.user.id)
  end
end

  
    
  
  2. Create a queries module for custom query logic:


defmodule MyApp.Blog.CommentQueries do
  import Ecto.Query

  def apply_query_option({:with_recent_activity, true}, query) do
    query
    |> where([c], c.inserted_at > ago(1, "day"))
  end

  def apply_query_option({:with_user_name, name}, query) do
    query
    |> join(:inner, [c], u in assoc(c, :user))
    |> where([_, u], u.name == ^name))
  end
end

  
    
  
  3. Use ContextKit.CRUD for basic contexts:


# Basic usage without scopes
defmodule MyApp.Blog do
  use ContextKit.CRUD,
    repo: MyApp.Repo,
    schema: MyApp.Blog.Comment,
    queries: MyApp.Blog.CommentQueries
end

  
    
  
  4. Or use ContextKit.CRUD.Scoped for Phoenix 1.8 Scopes and PubSub:


defmodule MyApp.Blog do
  use ContextKit.CRUD.Scoped,
    repo: MyApp.Repo,
    schema: MyApp.Blog.Comment,
    queries: MyApp.Blog.CommentQueries,
    pubsub: MyApp.PubSub,
    scope: Application.compile_env(:my_app, :scopes)[:user]
end
You can define custom query functions directly in your context module by setting queries: __MODULE__,
eliminating the need for a separate queries module. This is convenient for simpler contexts
where you don't need to share query logic.

  
    
  
  Features



  
    
  
  Standard CRUD Operations


ContextKit automatically generates common CRUD functions:
# Get query for use with Repo functions
Blog.query_comments()
Blog.query_comments(status: "published")

# New record
Blog.new_comment(%{status: "published"})

# List records with filtering and pagination
Blog.list_comments(status: "published", paginate: [page: 1, per_page: 20])

# Get single record
Blog.get_comment(123)
Blog.get_comment!(123)  # Raises if not found

# Get one record by criteria
Blog.one_comment(user_id: 1)
Blog.one_comment!(user_id: 1)  # Raises if not found

# Save records (insert or update based on record state)
Blog.save_comment(comment, %{body: "New or updated content"})
Blog.save_comment!(comment, %{body: "New or updated content"})  # Raises on error

# Create records
Blog.create_comment(%{body: "Great post!", user_id: 1})
Blog.create_comment!(%{body: "Great post!", user_id: 1})  # Raises on error

# Update records
Blog.update_comment(comment, %{body: "Updated comment"})
Blog.update_comment!(comment, %{body: "Updated comment"})  # Raises on error

# Get changeset
Blog.change_comment(comment)
Blog.change_comment(comment, %{body: "Changed comment"})

# Delete records
Blog.delete_comment(comment)
Blog.delete_comment(user_id: 1, body: "Specific comment")

  
    
  
  Scoped Operations (Phoenix 1.8)


When using ContextKit.CRUD.Scoped, all CRUD operations can take a scope parameter:
# Get query scoped to current user
Blog.query_comments(socket.assigns.current_scope)
Blog.query_comments(socket.assigns.current_scope, status: "published")

# List comments for the current user
Blog.list_comments(socket.assigns.current_scope)

# Get comment by ID only if it belongs to the current user
Blog.get_comment(socket.assigns.current_scope, 123)

# Save comment (insert or update) with scope
Blog.save_comment(socket.assigns.current_scope, comment, %{body: "New or updated content"})

# Create comment for the current user
Blog.create_comment(socket.assigns.current_scope, %{body: "Great post!"})

# Update comment only if it belongs to the current user
Blog.update_comment(socket.assigns.current_scope, comment, %{body: "Updated comment"})

# Delete comment only if it belongs to the current user
Blog.delete_comment(socket.assigns.current_scope, comment)

  
    
  
  Advanced Filtering


Supports a wide range of filter operations:
# Basic equality
Blog.list_comments(status: "published")

# Complex filters
Blog.list_comments(filters: [
  %{field: :body, op: :ilike, value: "%awesome%"},
  %{field: :status, op: :in, value: ["published", "pending"]},
  %{field: :inserted_at, op: :gt, value: ~N[2023-01-01 00:00:00]}
])
All fields from the schema can be filtered on automatically.
Any option not recognized as a field filter or standard query option is treated as a custom query option and passed to
the queries module's apply_query_option/2 function.

  
    
  
  Query Operations


ContextKit provides query functions that return Ecto queries without executing them,
useful for aggregation, composition, or further customization:
# Get a base query for all comments
query = Blog.query_comments()

# Apply filters to a query
query = Blog.query_comments(status: "published")

# Get a scoped query (with ContextKit.CRUD.Scoped)
query = Blog.query_comments(socket.assigns.current_scope)

# Use with Repo functions
MyApp.Repo.aggregate(query, :count)
MyApp.Repo.all(query)

# Compose with other queries
query
|> join(:left, [c], u in assoc(c, :user))
|> group_by([c, u], u.id)
|> select([c, u], {u.name, count(c.id)})
|> MyApp.Repo.all()
Query functions return an Ecto.Query which is perfect for:
	Computing aggregations (count, sum, etc.)
	Creating complex reports with multiple joins
	Building sub-queries
	Performing custom operations that ContextKit's standard functions don't cover


  
    
  
  Pagination


Built-in pagination support:
{comments, pagination} = Blog.list_comments(
  status: "published",
  paginate: [page: 2, per_page: 20]
)

# pagination struct includes:
# - total_count
# - total_pages
# - current_page
# - per_page
# - has_next_page?
# - has_previous_page?
# - next_page
# - previous_page

  
    
  
  PubSub Integration


When using ContextKit.CRUD.Scoped with pubsub and scope options, automatic broadcasting occurs for create, update, and delete operations:
# Subscribe to scoped updates
Blog.subscribe_comments(socket.assigns.current_scope)

# Now the current process will receive messages like:
# {:created, %Comment{}}
# {:updated, %Comment{}}
# {:deleted, %Comment{}}

# Broadcast custom messages
Blog.broadcast_comment(socket.assigns.current_scope, {:custom_event, comment})

  
    
  
  Custom Query Options


Extend with custom query logic:
# In your queries module
def apply_query_option({:with_recent_activity, true}, query) do
  query
  |> where([c], c.inserted_at > ago(1, "day"))
end

# Usage
Blog.list_comments(with_recent_activity: true)

  
    
  
  Configuration Options


When using ContextKit.CRUD or ContextKit.CRUD.Scoped, you can configure:
	repo: Your Ecto repository module
	schema: The Ecto schema module
	queries: Module containing custom query functions
	except: List of operations to exclude (:new, :list, :get, :one, :delete, :create, :update, :change, :subscribe, :broadcast)
	plural_resource_name: Custom plural name for list functions

Additional options for ContextKit.CRUD.Scoped:
	pubsub: Phoenix.PubSub module for real-time features
	scope: Configuration for Phoenix 1.8 scopes


  
    
  
  Best Practices


	Choose the right module: ContextKit.CRUD for simple contexts, ContextKit.CRUD.Scoped for contexts with scopes
	Create separate query modules for complex filtering logic
	Override generated functions when you need custom behavior
	Use pagination for large datasets
	When using scopes, implement schema-level changeset functions that accept a scope parameter
	Use PubSub for real-time updates in LiveView applications


      




  

    
ContextKit.CRUD 
    



      
The ContextKit.CRUD module provides a convenient way to generate standard CRUD (Create, Read, Update, Delete)
operations for your Ecto schemas. It reduces boilerplate code by automatically generating commonly used database
interaction functions.

  
    
  
  Setup


Add the following to your context module:
defmodule MyApp.Blog do
  use ContextKit.CRUD,
    repo: MyApp.Repo,
    schema: MyApp.Blog.Comment,
    queries: MyApp.Blog.CommentQueries,
    except: [:delete],                    # Optional: exclude specific operations
    plural_resource_name: "comments"      # Optional: customize plural name
end

  
    
  
  Required Options


	:repo - The Ecto repository module to use for database operations
	:schema - The Ecto schema module that defines your resource
	:queries - Module containing query-building functions for advanced filtering


  
    
  
  Optional Options


	:except - List of operation types to exclude (:list, :get, :one, :delete, :create, :update, :change)
	:plural_resource_name - Custom plural name for list functions (defaults to singular + "s")


  
    
  
  Generated Functions


For a schema named Comment, the following functions are generated:

  
    
  
  Query Operations


	query_comments/0 - Returns a base query for all comments
	query_comments/1 - Returns a filtered query based on options (without executing)


  
    
  
  New Operations


	new_comment/0 - Returns a new comment
	new_comment/1 - Returns a new comment with params
	new_comment/2 - Returns a new comment with params and opts


  
    
  
  List Operations


	list_comments/0 - Returns all comments
	list_comments/1 - Returns filtered comments based on options


  
    
  
  Get Operations


	get_comment/1 - Fetches a single comment by ID
	get_comment/2 - Fetches a comment by ID with additional filters
	get_comment!/1 - Like get_comment/1 but raises if not found
	get_comment!/2 - Like get_comment/2 but raises if not found


  
    
  
  Single Record Operations


	one_comment/1 - Fetches a single comment matching the criteria
	one_comment!/1 - Like one_comment/1 but raises if not found


  
    
  
  Save Operations


	save_comment/2 - Saves (inserts or updates) a comment with provided attributes
	save_comment!/2 - Like save_comment/2 but raises on invalid attributes


  
    
  
  Create Operations


	create_comment/1 - Creates a new comment with provided attributes
	create_comment!/1 - Like create_comment/1 but raises on invalid attributes


  
    
  
  Update Operations


	update_comment/2 - Updates comment with provided attributes
	update_comment!/2 - Like update_comment/2 but raises on invalid attributes


  
    
  
  Change Operations


	change_comment/2 - Returns a changeset for the comment with optional changes


  
    
  
  Delete Operations


	delete_comment/1 - Deletes a comment struct or by query criteria


  
    
  
  Query Options


All functions that accept options support:
	Basic filtering with field-value pairs
	Complex queries via Ecto.Query
	Pagination via paginate: true or paginate: [page: 1, per_page: 20]
	Custom query options defined in your queries module


  
    
  
  Examples


# Get a query for comments (for use with Repo.aggregate, etc.)
query = MyApp.Blog.query_comments(status: :published)
MyApp.Repo.aggregate(query, :count)

# New comment
MyApp.Blog.new_comment()

# List all comments
MyApp.Blog.list_comments()

# List published comments with pagination
MyApp.Blog.list_comments(status: :published, paginate: [page: 1])

# Get comment by ID with preloads
MyApp.Blog.get_comment(123, preload: [:user])

# Save a comment (insert if new, update if existing)
MyApp.Blog.save_comment(comment, %{body: "New or updated content"})

# Create a new comment
MyApp.Blog.create_comment(%{body: "Great post!", user_id: 1})

# Update a comment
MyApp.Blog.update_comment(comment, %{body: "Updated content"})

# Get a changeset for updates
MyApp.Blog.change_comment(comment, %{body: "Changed content"})

# Delete comment matching criteria
MyApp.Blog.delete_comment(body: "Specific content to delete")

  
    
  
  Queries Module


The required :queries module should implement apply_query_option/2, which receives a query option and the current query and returns a modified query. This allows for custom filtering, sorting, and other query modifications.
defmodule MyApp.Blog.CommentQueries do
  import Ecto.Query

  def apply_query_option({:with_user_name, name}, query) do
    query
    |> join(:inner, [c], u in assoc(c, :user))
    |> where([_, u], ilike(u.name, ^"%#{name}%"))
  end

  def apply_query_option({:recent_first, true}, query) do
    order_by(query, [c], desc: c.inserted_at)
  end

  def apply_query_option(_, query), do: query
end
Each generated function can be overridden in your context module if you need custom behavior.

      




  

    
ContextKit.CRUD.Scoped 
    



      
The ContextKit.CRUD.Scoped module provides a convenient way to generate standard CRUD (Create, Read, Update, Delete) operations for your Ecto schemas. It reduces boilerplate code by automatically generating commonly used database interaction functions.
Additionally, it supports scopes from Phoenix 1.8.

  
    
  
  Setup


Add the following to your context module:
defmodule MyApp.Blog do
  use ContextKit.CRUD.Scoped,
    repo: MyApp.Repo,
    schema: MyApp.Blog.Comment,
    queries: MyApp.Blog.CommentQueries,
    pubsub: MyApp.PubSub,                 # For realtime notifications via PubSub
    scope: Application.compile_env(:my_app, :scopes)[:user], # To gain support for Phoenix 1.8 scopes.
    except: [:delete],                    # Optional: exclude specific operations
    plural_resource_name: "comments"      # Optional: customize plural name
end

  
    
  
  Required Options


	:repo - The Ecto repository module to use for database operations
	:schema - The Ecto schema module that defines your resource
	:queries - Module containing query-building functions for advanced filtering
	:pubsub - The Phoenix.PubSub module to use for real-time features (required for subscription features)
	:scope - Configuration for scoping resources to specific contexts (e.g., user)


  
    
  
  Optional Options


	:except - List of operation types to exclude (:new, :list, :get, :one, :delete, :create, :update, :change, :subscribe, :broadcast)
	:plural_resource_name - Custom plural name for list functions (defaults to singular + "s")


  
    
  
  Generated Functions


For a schema named Comment, the following functions are generated:

  
    
  
  Query Operations


	query_comments/0 - Returns a base query for all comments
	query_comments/1 - Returns a filtered query based on options (without executing)
	query_comments/2 - Returns a scoped and filtered query if :scope is configured


  
    
  
  New Operations


	new_comment/0 - Returns a new comment
	new_comment/1 - Returns a new comment with params
	new_comment/2 - Returns a new comment with params and opts


  
    
  
  List Operations


	list_comments/0 - Returns all comments
	list_comments/1 - Returns filtered comments based on options
	list_comments/2 - Returns scoped and filtered comments if :scope is configured


  
    
  
  Get Operations


	get_comment/1 - Fetches a single comment by ID
	get_comment/2 - Fetches a comment by ID with additional filters
	get_comment/3 - Fetches a scoped comment by ID with additional filters if :scope is configured
	get_comment!/1 - Like get_comment/1 but raises if not found
	get_comment!/2 - Like get_comment/2 but raises if not found
	get_comment!/3 - Like get_comment/3 but raises if not found (with scope)


  
    
  
  Single Record Operations


	one_comment/1 - Fetches a single comment matching the criteria
	one_comment/2 - Fetches a scoped single comment if :scope is configured
	one_comment!/1 - Like one_comment/1 but raises if not found
	one_comment!/2 - Like one_comment/2 but raises if not found (with scope)


  
    
  
  Save Operations


	save_comment/1 - Saves (inserts or updates) a comment
	save_comment/2 - Saves a comment with the provided attributes
	save_comment/3 - Saves a scoped comment with attributes if :scope is configured
	save_comment!/1 - Like save_comment/1 but raises on invalid attributes
	save_comment!/2 - Like save_comment/2 but raises on invalid attributes
	save_comment!/3 - Like save_comment/3 but raises on invalid attributes (with scope)


  
    
  
  Create Operations


	create_comment/1 - Creates a new comment with provided attributes
	create_comment/2 - Creates a scoped comment if :scope is configured
	create_comment!/1 - Like create_comment/1 but raises on invalid attributes
	create_comment!/2 - Like create_comment/2 but raises on invalid attributes (with scope)


  
    
  
  Update Operations


	update_comment/2 - Updates comment with provided attributes
	update_comment/3 - Updates scoped comment if :scope is configured
	update_comment!/2 - Like update_comment/2 but raises on invalid attributes
	update_comment!/3 - Like update_comment/3 but raises on invalid attributes (with scope)


  
    
  
  Change Operations


	change_comment/1 - Returns a changeset for the comment
	change_comment/2 - Returns a changeset for the comment with changes
	change_comment/3 - Returns a changeset for the scoped comment with changes if :scope is configured


  
    
  
  Delete Operations


	delete_comment/1 - Deletes a comment struct or by query criteria
	delete_comment/2 - Deletes a scoped comment if :scope is configured


  
    
  
  PubSub Operations (if :pubsub and :scope are configured)


	subscribe_comments/1 - Subscribes to the scoped comments topic
	broadcast_comment/2 - Broadcasts a message to the scoped comments topic


  
    
  
  Query Options


All functions that accept options support:
	Basic filtering with field-value pairs
	Complex queries via Ecto.Query
	Pagination via paginate: true or paginate: [page: 1, per_page: 20]
	Custom query options defined in your queries module
	Scoping via scope when using scoped functions


  
    
  
  Examples


# Get a query for comments (for use with Repo.aggregate, etc.)
query = MyApp.Blog.query_comments(status: :published)
MyApp.Repo.aggregate(query, :count)

# Get a query for comments scoped to current user
query = MyApp.Blog.query_comments(socket.assigns.current_scope)
MyApp.Repo.aggregate(query, :count)

# Get a query for comments scoped to current user with additional filters
query = MyApp.Blog.query_comments(socket.assigns.current_scope, status: :published)
MyApp.Repo.aggregate(query, :count)

# New comment
MyApp.Blog.new_comment()

# List all comments
MyApp.Blog.list_comments()

# List all comments belonging to current user
MyApp.Blog.list_comments(socket.assigns.current_scope)

# List published comments with pagination
MyApp.Blog.list_comments(status: :published, paginate: [page: 1])

# Get comment by ID with preloads
MyApp.Blog.get_comment(123, preload: [:user])

# Get comment by ID that belongs to current user
MyApp.Blog.get_comment(socket.assigns.current_scope, 123)

# Save a new comment (will be inserted)
MyApp.Blog.save_comment(%Comment{}, %{body: "Great post!", user_id: 42})

# Save an existing comment (will be updated)
MyApp.Blog.save_comment(existing_comment, %{body: "Updated content"})

# Save a comment with scope (will insert or update depending on the record's state)
MyApp.Blog.save_comment(socket.assigns.current_scope, comment, %{body: "Content"})

# Save a comment or raise on errors
MyApp.Blog.save_comment!(comment, %{body: "Content that must be saved"})

# Create a new comment (manually specifying user_id)
MyApp.Blog.create_comment(%{body: "Great post!", user_id: 42})

# Create a new comment that automatically belongs to current user
MyApp.Blog.create_comment(socket.assigns.current_scope, %{body: "Great post!"})

# Update a comment
MyApp.Blog.update_comment(comment, %{body: "Updated content"})

# Update a comment that belongs to current user
MyApp.Blog.update_comment(socket.assigns.current_scope, comment, %{body: "Updated content"})

# Get a changeset for updates
MyApp.Blog.change_comment(comment, %{body: "Changed content"})

# Get a changeset for updates with scope
MyApp.Blog.change_comment(socket.assigns.current_scope, comment, %{body: "Changed content"})

# Delete comment
MyApp.Blog.delete_comment(comment)

# Delete comment that belongs to current user
MyApp.Blog.delete_comment(socket.assigns.current_scope, comment)

# Delete comment matching criteria
MyApp.Blog.delete_comment(body: "Specific content to delete")
Each generated function can be overridden in your context module if you need custom behavior.

  
    
  
  Queries Module


The required :queries module should implement apply_query_option/2, which receives a query option and the current query and returns a modified query. This allows for custom filtering, sorting, and other query modifications.
defmodule MyApp.Blog.CommentQueries do
  import Ecto.Query

  def apply_query_option({:with_user_name, name}, query) do
    query
    |> join(:inner, [c], u in assoc(c, :user))
    |> where([_, u], ilike(u.name, ^"%#{name}%"))
  end

  def apply_query_option({:recent_first, true}, query) do
    order_by(query, [c], desc: c.inserted_at)
  end

  def apply_query_option(_, query), do: query
end

  
    
  
  Scope


Read more about scopes.
A scope is a data structure used to keep information about the current request or session, such as the current user logged in, permissions, and so on. By using scopes, you have a single data structure that contains all relevant information, which is then passed around so all of your operations are properly scoped.
Usually you configure it with the same scope that you use in your Phoenix application:
scope: Application.compile_env(:my_app, :scopes)[:user],
pubsub: MyApp.PubSub # pubsub config is required for scopes
When a scope is configured, all relevant CRUD functions take an additional scope parameter as their first argument, ensuring that operations only affect records that belong to that scope.
For our comments example, the scope makes sure that comments are only listed, edited, or deleted by the user who created them.

  
    
  
  Scope Configuration


If you pass in the scope with Application.compile_env(:my_app, :scopes)[:user] - it will work automatically. You can also configure the scope manually. The scope configuration should be a keyword list with the following keys:
	:module - The module that defines the scope struct
	:access_path - Path to access the scoping value (e.g., [:user, :id] for user ID)
	:schema_key - The field in the schema that corresponds to the scope (e.g., :user_id)

For our comments example, this would typically be:
scope: [
  module: MyApp.Accounts.Scope,
  access_path: [:user, :id],
  schema_key: :user_id
]

  
    
  
  Subscription Example


# Subscribe to comments created, updated, or deleted by the current user
MyApp.Blog.subscribe_comments(socket.assigns.current_scope)

# Now the current process will receive messages like:
# {:created, %Comment{}}
# {:updated, %Comment{}}
# {:deleted, %Comment{}}

  
    
  
  Broadcasting Example


# Broadcast a custom message to all subscribers for current user's comments
MyApp.Blog.broadcast_comment(socket.assigns.current_scope, {:custom_event, comment})
Following messages are also broadcasted automatically for all create, update, delete operations:
{:created, %Comment{}}
{:updated, %Comment{}}
{:deleted, %Comment{}}

      




  

    
ContextKit.Paginator 
    



      
Handles pagination for database queries in ContextKit.
This module provides pagination functionality with configurable page size,
automatic limit/offset calculation, and metadata about the paginated results.

ContextKit.Query 
    



      
Provides dynamic query building functionality with extensive filtering operations for Ecto queries.
This module is core to ContextKit's filtering capabilities, providing a flexible way to build
complex database queries from runtime parameters.

  
    
  
  Query Building


The query building process consists of:
	Creating a base query with proper binding names
	Applying field filters based on schema fields
	Applying standard query options (order_by, limit, etc.)
	Applying custom query options


  
    
  
  Field Filters


Supports an extensive set of filter operations through a standardized filter syntax:
%{field: field_name, op: operator, value: value}

  
    
  
  Available Operators


Basic Comparison:
	:== - Equality
	:!= - Inequality
	:< - Less than
	:<= - Less than or equal
	:> - Greater than
	:>= - Greater than or equal

Null Checks:
	:empty - Check if field is NULL
	:not_empty - Check if field is NOT NULL

List Operations:
	:in - Check if field value is in list
	:not_in - Check if field value is not in list
	:contains - Check if list field contains value
	:not_contains - Check if list field does not contain value

Pattern Matching:
	:like - Case-sensitive pattern matching
	:not_like - Negative case-sensitive pattern matching
	:ilike - Case-insensitive pattern matching
	:not_ilike - Negative case-insensitive pattern matching
	:=~ - Shorthand for :ilike

Multiple Pattern Matching:
	:like_and - All patterns must match (case-sensitive)
	:like_or - Any pattern must match (case-sensitive)
	:ilike_and - All patterns must match (case-insensitive)
	:ilike_or - Any pattern must match (case-insensitive)


  
    
  
  Standard Query Options


  The following standard query options are supported:
	:order_by - Specify sort order
	:group_by - Group results
	:limit - Limit number of results
	:preload - Preload associations
	:paginate - Enable pagination with optional configuration
	:select - Select only specific fields from the database

